U.S. DEPARTMENT OF COMMERCE
NATIONAL OCEANIC AND ATMOSPHERIC ADMINISTRATION
NATIONAL WEATHER SERVICE
NATIONAL METEOROLOGICAL CENTER

OFFICE NOTE 263

Spectral Model Precipitation Algorithms

Joseph P. Gerrity, Jr.
Robert Hirano
Development Division

AUGUST 1982

1.0 Introduction

In this note, we review the algorithms for computing convective and large scale moist adiabatic processes in the spectral model. The methods have been modified from those used prior to August 1982. The modifications are directed primarily toward providing a thermodynamically consistent treatment rather than toward producing empirically optimal predicted rainfall amounts. The latter goal doesn't seem tractable at the present stage of development of the global, spectral model.

The paper describes the convective scale parameterization in section 2.0, the large scale parameterization in section 3.0 and concludes with a summary of recent tests of the method in section 4.0.

2.0 The Moist-Convection Algorithm

At each time step, after computing the response of the wind, mass and water vapor fields to dry-adiabatic processes, the fields are assembled on the Gaussian grid. The first-process, of a diabatic nature, now considered is the requirement for moist adiabatic, convective adjustment of the newly predicted temperature and water vapor variables. The process is simulated using a method based upon Kuo's idealization of the stabilizing action of sub-grid scale cumulus convection, therefore, the subroutine is termed "CONKUO".

The variables passed to and from the subroutine are

DT	the time step (sec)
SATCRI	a fraction (no longer used)
ACUMT	a negative constant presently -1.E-7
NKUO	a counter used to record the number of points at which the algorithm is fully exercised
PS	the predicted surface pressure (cb)

DEL a array used in defining the pressure interval across each model layer (DP = DEL*PS).

SL an array to be transferred to the subroutine which computes the moist adiabat (MSTADB), it also defines the pressure at the mid-point of model layers PRESS = SL*PS.

QN an array containing the specific humidity variable at the previous time step.

QN1 an array containing the specific humidity predicted for the present time.

TN1 an array containing the temperature (°K) predicted for the present time.

ACUM an array used in summing up the water vapor convergence over the past time step.

DQ an array used to hold the water vapor change, QN1-QN, which is computed in this routine.

GESHEM an array used hold the accumulated precipitation at a grid point.

RELKUO a constant 0.8 (no longer used).

MSTA a counter used to record points at which a problem is encountered in computing the moist adiabat.

The first significant computation in the subroutine involves the calculation of DQ, the difference between QN1 and QN. It measures the change of the specific humidity in the time interval DT.

One next computes, the integral

$$ACUM = \qquad DQ \cdot dp \qquad -1.E-8 \quad (\text{cb sec}^{-1}))$$

over the four lowest layers of the model.

Two "if - tests" are next made. If ACUM, just calculated, is negative, or if DQ in the lowest layer is negative, then the decision is made to by-pass any moist convective adjustment at the grid point.

Two additional "if - tests" are made on the temperature at this point: if the temperature is less than 5° celsius, or if a temperature inversion exists between the lowest two layers then the decision is made to by-pass the remainder of the calculation at the gridpoint in question.

The next step is to compute the saturation value of specific humidity in the lowest layer. Then, the forecast relative humidity in the lowest layer is computed and, if it is less than 65%, the decision is made to bypass further calculation at the gridpoint in question.

Having reached this point in the subroutine, we know that the following conditions are true

1. Moisture convergence occurred in the model's lowest layer during the previous time step.
2. The lowest layer is warm, T>5°C, and moist, RH>65%.
3. The temperature lapse rate between the lowest two layers is more unstable than isothermal.

The next step in the subroutine is the contruction of the sounding associated with the moist adiabat passing through the midpoint of the lowest layer at a temperature T given by

$$T = 0.75\ T_1 + 0.25\ T_2$$

where T_1 is the lowest layer's temperature and T_2 is the temperature in the second lowest layer. The calculation is done in subroutine 'MSTADB' which also returns the index KTOP of the highest model layer for which the moist adiabat's temperature exceeds the forecast temperatures. Actually, one layer in which the forecast temperature exceeds the moist adiabat's temperature may intervene between the lowest layer and KTOP.

Should KTOP be less than 4, that is, if the 'cloud' is not sufficiently deep, the subroutine ignores further computation at the grid point in question. Another "if test" rules out the existence of layer, between KTOP and the first layer, in which the cloud is colder than the forecast temperature.

At this point in the subroutine, we are assured that the moist adiabat representing the sub-grid scale cloud, extends upward through at least three layers above the model's first layer, in such a way that the cloud is warmer than the forecast temperatures.

We now use a modification of Kuo's method to calculate adjustments to the forecast temperature and humidity within the layers that are presumably influenced by deep, sub-grid scale, cumulus clouds.

An available water parameter is formed by adding together the water vapor which has converged in the lowest two layers during the previous time step. This must be a positive quantity to allow an adjustment to occur.

One then forms the amount of water which must be available to accomplish a complete adjustment of the forecast temperature and humidity to the values associated with the cloud values as determined from the moist adiabat.

The ratio of the available water to that which is required for total adjustment defines a fraction QEFF. To avoid accumulating very small adjustments it is required that QEFF exceed 0.001, and, of course, QEFF cannot be permitted to exceed unity.

The forecast temperature and humidity in layers between 2 and KTOP are then adjusted toward the moist adiabat values by setting adjusted values T* and q* according to the relations

$$T^* = (1.-QEFF)TN1 + QEFF\,(TMST)$$

$$q^* = (1.-QEFF)QN1 + QEFF(QMST)$$

where TMST and QMST are the temperature and saturation specific humidity values associated with the moist adiabat (i.e. cloud) in the appropriate layers.

The specific humidity in the lowest two layers is reduced to account for the upward transfer of the available water in building the cloud. The temperature in the lowest layer is unaffected by this process.

Finally an estimate of convective rainfall in computed from the water used to produce the temperature adjustments. Since this rainfall is associated with a moisture convergence over one time step the full amount of convective rain is added to the storage location, GESHEM.

3.0 The Large Scale Condensation Algorithm

The basic method for computing the condensation and precipitation of water vapor is unchanged, but a number of refinements have been introduced which provide greater thermodynamic consistency in the calculations. The principal changes are:

- Saturation is determined from the predicted temperature and pressure rather than through the use of an empirical saturation criterion. Furthermore, the initial specific humidity is no longer reduced. The lowest, or boundary, layer remains an exception at this time. Within that layer the initial specific humidity is reduced by the factor 0.9. Evaporation from the sea or from falling rain is not permitted to raise the lowest layer's relative humidity above 90%. These special treatments of the lowest layer are intended to suppress the formation of precipitation in that layer.
- The condensation adjustment accounts for the modification of saturation specific humidity as the latent heat raises the temperature. This wet-bulb approximation insures that, after the condensation process has been computed, the temperature and specific humidity are in agreement with the saturation condition. Previous methods allowed the existence of slightly subsaturated conditions after the condensation process was computed.

- Because the dry adiabatic processes that led to the existence of supersaturation occurred over a two time-step interval (leapfrog time integration) the accumulated precipitation reaching the ground is multiplied 0.5 before inserting it into the model's parameter for accumulating net precipitation. Previously, the full amount was entered which led to an overestimate of large scale precipitation being produced by the model. Since the accumulated precipitation is the only direct indicator of the location and intensity of latent heat release, the correction of this output will permit a more correct assessment of the impact of large-scale condensation on the circulation forecasts.

The parameters passed to and from the subroutine LRGSCL are

SATCRI	a parameter no longer needed
GESHEM	the array storing accumulated precipitation
TF	predicted temperature (°K)
QS	An array to hold the saturation specific humidity
QF	predicted specific humidity
PS	predicted surface pressure (centibars)
SATC	an array which is no longer needed
DEL	an array used in defining the pressure interval across each model layer, DP=DEL*PS
PREC	an array used to hold falling precipitation
SL	an array used to define the pressure at the mid-point of model layers. PRESS=SL*PS
B	an array which is no longer needed
SUPER	an array used to hold the amount of water associated with the condensation process.
DPOVG	an array used to scale specific humidity into precipitable water within a layer. It is also used as a temporary storage array in computing the saturation specific humidity.

The initial computation performed in the subroutine determines the saturation specific humidity in the several moisture bearing layers of the model. The precipitable water scaling array DPOVG is then defined, and the falling precipitation storage array PREC is initialized to zero.

A set of factors FACT, used to simulate the impact of drop-size on the effective rate of evaporation of falling rain, is defined.

The basic algorithm is now begun and works downward from the uppermost layer. The first computation evaluates SUPER, the amount of water deficit or excess with respect to saturation in a layer. If an excess is present (SUPER>0.), the wet-bulb approximation is used to adjust the predicted temperature and specific humidity to an equilibrium saturated state. The amount of water condensed is used to reduce the predicted specific humidity and the predicted temperature is raised to account for the release of latent heat. The parameter PREC is augmented by the condensed water, so that PREC contains the amount of water falling as precipitation downward through the column.

If the layer is not supersaturated, i.e. if SUPER is negative, the algorithm checks to see if there is falling precipitation present and if evaporation into the layer is appropriate. Evaporation of falling rain is permitted only if the layer's relative humidity is less than 80%. An exception is that in the lowest layer evaporation of falling precipitation is permitted as long as the layer relative humidity is less than 90%.

If evaporation is possible, the deficit, i.e. the difference between actual water content and the appropriate upper bound, is computed as EVAP. This quantity is scaled to a potential evaporation POTEVP amount by multiplication by the factor FACT. The amount of water for evaporation is determined to be either POTEVP or the amount of falling rain PREC whichever is less.

The layer's specific humidity is increased and the temperature decreased to account for the evaporation. The wetbulb adjustment approximation is once more used so that the adjusted state is consistent with the thermodynamic process. Finally, the parameter PREC is adjusted to account for the reduction of falling rain brought about by the evaporation process.

Having completed consideration of a layer the algorithm passes to the next lower layer and repeats the logic. When all layers have been treated the remaining PREC, if any, is multiplied by 0.5 and added to GESHEM, the storage location for accumulated precipitation.

4.0 Test results.

The new method has been used to produce forecasts for five cases; two cases were integrated to 84 hrs, one to 60 hrs and two others to 48 hrs. During the conduct of the tests small modifications were introduced in the algorithms in order to improve the logic of the programs. The changes do not affect the basic character of the results in the writers' opinion.

In the table below we present certain statistics for each 12 hour forecast. The data were computed using the output precipitation forecasts on the 65x65 polar stereographic map of gridpoints. The model itself uses a Gaussian grid of latitude and longitude, so the basic data were first interpolated to the Cartesian map array as is ordinary operational practice.

For each 12 hour period, the amount of precipitation (inches) output on the gridpoints was calculated, as was the number of gridpoints at which the precipitation was forecast to equal or exceed 0.01 inches. These two statistics are indicated under AMT and No. respectively. Also shown is the average (over points with precipitation) 12 hour accumulation AVE and the fraction of the 4225 possible points at which precipitation was forecast to occur.

For each case, we show the results obtained with the operational spectral model (OLD) and with the modified method (NEW).

The average forecast amount of 12 hourly precipitation at points with precipitation is reduced from 0.227 "for the "OLD" method to 0.136" using the "NEW" method. This result is consistent with the change in the LRGSCL algorithm to correct for use of the leapfrog integration method.

The average faction of the 4225 grid points predicted to have precipitation is increased form 6.7% under the OLD method to 9.2% using the NEW method. Most of this increase is accounted for by the first 36 hours of the forecast. We may attribute the NEW results to the suppression of the previously used method for scaling down the intial specific humidity by the factor 0.9.

While not evident from the summary statistics, it is the opinion of the writers that the NEW method succeeds in accomplishing two synoptically useful results. First, the precipitation patterns associated with contiguous short wave perturbations appear to be separated more reasonably in the NEW method. Second, the OLD method's tendency to "hang-back" precipitation behind oceanic cold fronts has been significantly improved. This effect is primarily produced by the adjustment to the rate of evaporation and sensible heat transfer from the ocean surface which has been incorporated into the modified program. The new code allows these processes to function at only 50% of the previously employed rate.

There are no verification scores quoted here because most of the integration domain is devoid of verifying observations and secondly only a small number of cases have been integrated. We don't anticipate a significant change in the general level of forecast skill to result from the modified algorithms, but we do hope that the forecasts will prove more useful in diagnosing situations in which large amounts of latent heat may be producing changes in the circulation. In the five test cases, we did not detect significant differences in the circulation predictions produced by the model.

Case 12Z 05 January 1982

	OLD				NEW			
HR	AMT	No.	Ave.	FRACT(%)	AMT	No.	Ave.	FRACT(%)
12	50	230	0.22	5.4	40	340	0.12	8.0
24	73	340	0.21	8.0	60	380	0.16	9.0
36	102	400	0.26	9.5	70	360	0.19	8.5
48	103	410	0.25	9.7	73	370	0.20	8.8

Case 00Z 17 June 1982

	OLD				NEW			
HR	AMT	No.	Ave.	FRACT(%)	AMT	No.	Ave.	FRACT(%)
12	18	120	0.15	2.8	15	220	0.07	5.2
24	52	160	0.32	3.8	45	270	0.17	6.4
36	71	210	0.34	5.0	62	282	0.22	6.7
48	70	215	0.32	5.1	73	285	0.26	6.7

Case 12Z 01 April 1982

	OLD				NEW			
HR	AMT	No.	Ave.	FRACT(%)	AMT	No.	Ave.	FRACT(%)
12	18	150	.12	3.6	71	510	.14	12.1
24	54	230	.23	5.4	75	570	.13	13.5
36	90	310	.29	7.3	53	500	.11	11.8
48	100	360	.28	8.5	52	430	.12	10.2
60	87	380	.23	9.0	47	450	.10	10.6

Case 00Z 20 March 1982

	OLD				NEW			
HR	AMT	No.	Ave.	FRACT(%)	AMT	No.	Ave.	FRACT(%)
12	26	213	.12	5.0	39	410	.10	9.7
24	43	250	.17	5.9	43	470	.09	11.1
36	62	300	.20	7.1	58	480	.12	11.4
48	103	340	.30	8.0	105	470	.22	11.1
60	124	500	.25	11.8	64	570	.11	13.5
72	80	440	.18	10.4	36	380	.09	9.0
84	56	390	.14	9.2	34	390	.09	9.2

Case 00Z 20 March 1982

	OLD				NEW			
HR	AMT	No.	Ave.	FRACT(%)	AMT	No.	Ave.	FRACT(%)
12	17	150	.11	3.6	26	315	.08	7.5
24	38	210	.18	5.0	35	430	.08	10.2
36	60	250	.24	5.9	40	375	.11	8.9
48	70	300	.23	7.1	47	420	.11	9.9
60	67	350	.19	8.3	45	450	.10	10.6
72	62	325	.19	7.7	37	405	.09	9.6
84	64	360	.18	8.5	35	370	.09	8.8

TABLE SUMMARY STATISTICS FROM TESTS SEE TEXT FOR DETAILS.

Acknowledgements

We appreciate the technical support provided by Drs. Sela and Stackpole in the conduct of the experimental integrations.

We also thank Harlan Saylor and David Olson for reviewing the forecast charts and providing comments.

```
      SUBROUTINE CONKUO(DT,SATCRI,ACUMT,NKUO,PS,DEL,SL,QN,QN1,TN1,
     1 ACUM,DQ,GESHEM,RELKUO,MSTA)
      DIMENSION GESHEM(#LONF)
      DIMENSION PS(#LONF),DEL(#LEVS),SL(#LEVS),
     1QN(#LONF,#LEVH),QN1(#LONF,#LEVH),DQ(#LONF,#LEVH),
     2TN1(#LONF,#LEVS)
      DIMENSION ACUM(#LONF),PRESS(#LEVH),TIN(#LEVH),QIN(#LEVH),
     1TMST(#LEVH),QMST(#LEVH),DTKUO(#LEVH),DQKUO(#LEVH),ESAT(#LEVH)
      IF(DT.EQ.0.) RETURN
C***
C*** DEL=DEL SIGMA (DEL P OVR PSFC)
C*** PS=SFC PRES (CB),  RELKUO(=.80) IS THE SATURATION CRITERION
C*** ACMT = TEN TO MINUS SEVEN PER SEC
      RDT=1./DT
      CPOVL=#CP/2.5E+6
      RELEPS=RELKUO*0.622
      DO 5 K=1,#LEVH
      DO 5 I=1,#LONF
5     DQ(I,K)=QN1(I,K)-QN(I,K)
C*** NET TIME RATE MOISTURE INFLOW ..LWST KACUM.. LYRS
      KACUM=4
      DO  10 I=1,#LONF
10    ACUM(I)=ACUMT*0.1
      DO 15 K=1,KACUM
      DO 15 I=1,#LONF
15    ACUM(I)=ACUM(I)+RDT*DQ(I,K)*DEL(K)
C***
C*** ENTER LON LOOP........
      DO 125 I=1,#LONF
C***
C*** TEST 1....SUFFICIENT WATER ACCUM? NET INFLO INTO LWST LYR?
      IF(ACUM(I).LT.0.) GO TO 125
      IF(DQ(I,1) .LE. .0) GO TO 125
      DO 12345 K=1,#LEVH
12345 PRESS(K)=SL(K)*PS(I)
C*** TIN, QIN ARE PRELIM TMP AND Q AT N+1, ZERO NEGATIVE Q
      DO 103 K=1,#LEVH
      TIN(K)=TN1(I,K)
      QIN(K)=QN1(I,K)
      IF(QN1(I,K) .LT. .0) QIN(K) = .0
103   CONTINUE
       IF(TIN(1) .LT. 278.16) GO TO 125
       IF(TIN(2) .GT. TIN(1)) GO TO 125
       CALL SATVAP(TIN,ESAT,2)
       QSATK = .65*.622*ESAT(1)/(PRESS(1)-0.378*ESAT(1))
       QDEF = QIN(1) - QSATK
C*** LWST LYR MIN  TMP AND RH CONDS PLUS NO TMP INVERSION EXISTS
       IF(QDEF .LT. .0) GO TO 125
C*** MODIFY TMP FOR MSTADB (SUBR PERMITS AN INTERVENING STBL LYR)
      TIN(1) =  TIN(1) + .25* (TIN(2) - TIN(1))
      HNEW=PS(I)/100.
      CALL MSTADB(HNEW, SL   ,TIN,1,#LEVH,THE,TMST,QMST,KTOP,IER)
C***  TMST AND QMST,K=1...KTOP CONTAIN CLOUD TEMPS AND SPECIFIC HUMID.
C**** ------------------------------
      IF(IER)105,106,105
105   MSTA=MSTA+1
      GO TO 125
106   CONTINUE
C**** ------------------------------
```

```
C**** ------------------------------
C***
C*** TEST 2...INSTABILITY.  KTOP .GE. 4,  NO STBL LYRS PERMITTED
         IF(KTOP.LT.4) GO TO 125
         DO 150 K = 2,KTOP
         X = TMST(K) - TIN(K)
         IF(X .LT. 0.) GO TO 125
150      CONTINUE
C*** HERE  KUO REQUIREMENTS ARE MET, FIRST COMPUTE WATER=TONS WATER
C*** SUBSTANCE  ACCUMULATED IN ONE TIME STEP PER SQ M IN LAYERS 2-KTOP
        NKUO=NKUO+1
C*** AVL  WATER  (WATER=WATER*G/PS) .GT. .0, COMES FM TWO LWST LYRS
        WATER=0.
        DO 107 K=1,2
        WATER=WATER+   DQ(I,K)*DEL(K)
107     CONTINUE
        IF(WATER .LE. .0) GO TO 125
C***
        Q1=0.
        Q2=0.
          DO 111 K=2,KTOP
          X=QMST(K)-QIN(K)
          IF(X)109,109,108
108       Q1=Q1+X*DEL(K)
109       X=TMST(K)-TIN(K)
          IF(X)111,111,110
110       Q2=Q2+X*DEL(K)
111       CONTINUE
C*** CONTRIB. FACTOR (FRAC CVRG?) .GE. MIN, NOT .GT. 1.0
        Q2=Q2*CPOVL
        QEFF=WATER/(Q1+Q2)
        IF(QEFF .LT. .001) GO TO 125
        IF(QEFF .GT. 1.00) QEFF = 1.00
C***
C*** CONVECTIVE  CONTRIBUTION TO HEATING AND MOISTENING
          DO 304 K=2,KTOP
          X=QEFF*(TMST(K)-TIN(K))
C***    CHANGE T ONLY IF X GT.0
          IF(X)300,300,301
300       X=0.
301       DTKUO(K)=X
          X=QEFF*(QMST(K)-QIN(K))
C***    CHANGE Q ONLY IF X GT.0
          IF(X)302,302,303
302       X=0.
303       DQKUO(K)=X
304       CONTINUE
        DTKUO(1)=0.
        DQKUO(1)=0.
        CEVAP=0.
          DO 305 K=2,KTOP
          TIN(K)=TIN(K)+DTKUO(K)
          QIN(K)=DQKUO(K)    + QIN(K)
          CEVAP=CEVAP+DQKUO(K)*DEL(K)
305       CONTINUE
C***
C*** CONVECTIVE PCPN  = AVL WATER, REMOVE WATER FM TWO LWST LYRS
        FALL = WATER - CEVAP
        GESHEM(I)=GESHEM(I)+PS(I)*FALL/9.8
         QIN(1) = QIN(1) - DQ(I,1)
         IF(DQ(I,2) .GT. 0.) QIN(2) = QIN(2)-DQ(I,2)
C***    RESTORE BASIC FIELDS
        TIN(1) = TN1(I,1)
        DO 310 K=1,KTOP
        QN1(I,K)=QIN(K)
        TN1(I,K)=TIN(K)
310     CONTINUE
125     CONTINUE
C***    END LON LOOP
C***
  % INCLUDE YSCHECK;
        RETURN
        END
```

```
      SUBROUTINE LRGSCL(SATCRI,GESHEM,      TF,QS,QF,PS,SATC,DEL,PREC,
     1 SL,B,SUPER,DPOVG)
C***  QF=Q(N+1),QS=SAT. Q AT T(N+1)=TF, PS=SURF. PRESS(CB)
C***
      DIMENSION GESHEM(#LONF),TF(#LONF,#LEVS),
     1QF(#LONF,#LEVH),QS(#LONF,#LEVH),PS(#LONF),
     2SATC(#LONF,#LEVH),DEL(#LEVS),PREC(#LONF), SL(#LEVS),
     3DPOVG(#LONF,#LEVH),SUPER(#LONF),B(#LONF),FACT(#LEVH)
      RLOCP=2.5E+6/#CP
      NVAL=#LONF*#LEVH
      RLRV=-2.51E+10/(1.61*2.87E+6)
      CONST=6.11*EXP(-RLRV/273.16) / 10.
      C1=CONST*  0.622
      C2=CONST*(-0.378)
      DO 10 K=1,#LEVH
      DO  9 I=1,#LONF
      DPOVG(I,K)=RLRV/TF(I,K)
9     CONTINUE
10    CONTINUE
      DO 12 K=1,#LEVH
      DO 11 I=1,#LONF
      QS(I,K) = EXP(DPOVG(I,K))
11    CONTINUE
12    CONTINUE
      DO 14 K=1,#LEVH
      DO 13 I=1,#LONF
      QS(I,K)=C1*QS(I,K)/(SL(K)*PS(I) + C2*QS(I,K))
13    CONTINUE
14    CONTINUE
      DO 2 K=1,#LEVH
      DO 1 I=1,#LONF
      DPOVG(I,K)=DEL(K)*PS(I)/#GRAV
1     CONTINUE
2     CONTINUE
      CALL XSTORE(PREC(1),0.0,#LONF)
      CALL XSTORE(SATC,1.0,#LONF*#LEVH)
C*** SET EVAPORATION FACTOR
      FACT(1)=.25
      FACT(2)=.16
      FACT(3)=.09
      DO 22 K=4,#LEVH
      FACT(K)=.04
22    CONTINUE
```

```
C*** PCPN PROCESS.....TOP LYR DOWNWARD
      RHSAT = .80
      DO 2190 K=1,#LEVH
      L=#LEVH+1-K
      IF(L.EQ.1)RHSAT=.90
      DO 2110 I=1,#LONF
      SUPER(I)=(QF(I,L)-QS(I,L))*DPOVG(I,L)
      B(I)=PREC(I)
2110  CONTINUE
      DO 3000 I=1,#LONF
      IF (SUPER(I).LE.0.)  GO TO 2120
C     HERE IF SUPER GT 0., COMPUTE WET-BULB
C     ADJUSTMENT TO T AND Q, AND AUGMENT
C     PRECIPITATION FALLING THROUGH COLUMN.
      ALPHA = -RLRV/(TF(I,L)**2)
      BETA = 1./RLOCP + ALPHA*QS(I,L)
      QCOND = (SUPER(I)/DPOVG(I,L)) * (1./(RLOCP*BETA))
      TF(I,L) = TF(I,L) + RLOCP*QCOND
      QF(I,L) = QF(I,L) - QCOND
      PREC(I) = PREC(I) + QCOND*DPOVG(I,L)
C     FINISHED WITH SUPER-SATURATED POINT
      GO TO 3000
2120  CONTINUE
C     HERE IF POINT IS NOT SUPER SATURATED
C     CHECK IF FALLING PRECIPITATION IS
C     PRESENT AND IF IT SHOULD BE
C     EVAPORATED.
      IF (PREC(I).LE.0.)  GO TO 3000
      EVAP = (QF(I,L)-QS(I,L)*RHSAT) * DPOVG(I,L)
      IF (EVAP.GE.0.)  GO TO 3000
      EVAP = -EVAP
C     HERE IF EVAPORATION OF FALLING
C     PRECIPITATION IS TO BE COMPUTED
C     COMPUTE POTENTIAL EVAPORATION
      POTEVP = FACT(L)*EVAP
C     ASK IF POTEVP EXCEEDS AVAILABLE
C     FALLING PRECIPITATION, COMPUTE
C     AMOUNT OF WATER EVAPORATED.
      EXCESS = PREC(I) - POTEVP
      AMTEVP = POTEVP
      IF (EXCESS.LT.0.)  AMTEVP=PREC(I)
C     ADJUST T AND Q TO ACCOUNT
C     FOR EVAPORATION, MODIFY PREC.
      ALPHA = -RLRV/(TF(I,L)**2)
      BETA = 1./RLOCP + ALPHA*QS(I,L)
      QEVAP = (AMTEVP/DPOVG(I,L)) * (1./(RLOCP*BETA))
      TF(I,L) = TF(I,L) - RLOCP*QEVAP
      QF(I,L) = QF(I,L) + QEVAP
      PREC(I) = PREC(I) - QEVAP*DPOVG(I,L)
      PREC(I) = AMAX1(PREC(I),0.)
C     FINISHED WITH POINT AT WHICH
C     FALLING PRECIP IS EVAPORATED
      GO TO 3000
3000  CONTINUE
2190  CONTINUE
C*** PCPN REACHES GROUND LEVEL....FACTOR OF .5 SINCE PCPN IS FOR
C*** TWO (LEAPFROG) TIME-STEPS....ACCUM PCPN (GESHEM) IS NOT LEAPFROGGED
      DO 20 I=1,#LONF
      GESHEM(I)=GESHEM(I)+PREC(I)*.5
   20 CONTINUE
 % INCLUDE YSCHECK;
      RETURN
      END
```